// File: @openzeppelin/contracts/math/SafeMath.sol

// SPDX-License-Identifier: MIT

pragma solidity ^0.7.0;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");

        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return sub(a, b, "SafeMath: subtraction overflow");
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");

        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return div(a, b, "SafeMath: division by zero");
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return mod(a, b, "SafeMath: modulo by zero");
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts with custom message when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b != 0, errorMessage);
        return a % b;
    }
}

// File: @openzeppelin/contracts/GSN/Context.sol



pragma solidity ^0.7.0;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



pragma solidity ^0.7.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: value }(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data, string memory errorMessage) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(bool success, bytes memory returndata, string memory errorMessage) private pure returns(bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/EnumerableSet.sol



pragma solidity ^0.7.0;

/**
 * @dev Library for managing
 * https://en.wikipedia.org/wiki/Set_(abstract_data_type)[sets] of primitive
 * types.
 *
 * Sets have the following properties:
 *
 * - Elements are added, removed, and checked for existence in constant time
 * (O(1)).
 * - Elements are enumerated in O(n). No guarantees are made on the ordering.
 *
 * ```
 * contract Example {
 *     // Add the library methods
 *     using EnumerableSet for EnumerableSet.AddressSet;
 *
 *     // Declare a set state variable
 *     EnumerableSet.AddressSet private mySet;
 * }
 * ```
 *
 * As of v3.3.0, sets of type `bytes32` (`Bytes32Set`), `address` (`AddressSet`)
 * and `uint256` (`UintSet`) are supported.
 */
library EnumerableSet {
    // To implement this library for multiple types with as little code
    // repetition as possible, we write it in terms of a generic Set type with
    // bytes32 values.
    // The Set implementation uses private functions, and user-facing
    // implementations (such as AddressSet) are just wrappers around the
    // underlying Set.
    // This means that we can only create new EnumerableSets for types that fit
    // in bytes32.

    struct Set {
        // Storage of set values
        bytes32[] _values;

        // Position of the value in the `values` array, plus 1 because index 0
        // means a value is not in the set.
        mapping (bytes32 => uint256) _indexes;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function _add(Set storage set, bytes32 value) private returns (bool) {
        if (!_contains(set, value)) {
            set._values.push(value);
            // The value is stored at length-1, but we add 1 to all indexes
            // and use 0 as a sentinel value
            set._indexes[value] = set._values.length;
            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function _remove(Set storage set, bytes32 value) private returns (bool) {
        // We read and store the value's index to prevent multiple reads from the same storage slot
        uint256 valueIndex = set._indexes[value];

        if (valueIndex != 0) { // Equivalent to contains(set, value)
            // To delete an element from the _values array in O(1), we swap the element to delete with the last one in
            // the array, and then remove the last element (sometimes called as 'swap and pop').
            // This modifies the order of the array, as noted in {at}.

            uint256 toDeleteIndex = valueIndex - 1;
            uint256 lastIndex = set._values.length - 1;

            // When the value to delete is the last one, the swap operation is unnecessary. However, since this occurs
            // so rarely, we still do the swap anyway to avoid the gas cost of adding an 'if' statement.

            bytes32 lastvalue = set._values[lastIndex];

            // Move the last value to the index where the value to delete is
            set._values[toDeleteIndex] = lastvalue;
            // Update the index for the moved value
            set._indexes[lastvalue] = toDeleteIndex + 1; // All indexes are 1-based

            // Delete the slot where the moved value was stored
            set._values.pop();

            // Delete the index for the deleted slot
            delete set._indexes[value];

            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function _contains(Set storage set, bytes32 value) private view returns (bool) {
        return set._indexes[value] != 0;
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function _length(Set storage set) private view returns (uint256) {
        return set._values.length;
    }

   /**
    * @dev Returns the value stored at position `index` in the set. O(1).
    *
    * Note that there are no guarantees on the ordering of values inside the
    * array, and it may change when more values are added or removed.
    *
    * Requirements:
    *
    * - `index` must be strictly less than {length}.
    */
    function _at(Set storage set, uint256 index) private view returns (bytes32) {
        require(set._values.length > index, "EnumerableSet: index out of bounds");
        return set._values[index];
    }

    // Bytes32Set

    struct Bytes32Set {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _add(set._inner, value);
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _remove(set._inner, value);
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(Bytes32Set storage set, bytes32 value) internal view returns (bool) {
        return _contains(set._inner, value);
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(Bytes32Set storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

   /**
    * @dev Returns the value stored at position `index` in the set. O(1).
    *
    * Note that there are no guarantees on the ordering of values inside the
    * array, and it may change when more values are added or removed.
    *
    * Requirements:
    *
    * - `index` must be strictly less than {length}.
    */
    function at(Bytes32Set storage set, uint256 index) internal view returns (bytes32) {
        return _at(set._inner, index);
    }

    // AddressSet

    struct AddressSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(AddressSet storage set, address value) internal returns (bool) {
        return _add(set._inner, bytes32(uint256(value)));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(AddressSet storage set, address value) internal returns (bool) {
        return _remove(set._inner, bytes32(uint256(value)));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(AddressSet storage set, address value) internal view returns (bool) {
        return _contains(set._inner, bytes32(uint256(value)));
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(AddressSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

   /**
    * @dev Returns the value stored at position `index` in the set. O(1).
    *
    * Note that there are no guarantees on the ordering of values inside the
    * array, and it may change when more values are added or removed.
    *
    * Requirements:
    *
    * - `index` must be strictly less than {length}.
    */
    function at(AddressSet storage set, uint256 index) internal view returns (address) {
        return address(uint256(_at(set._inner, index)));
    }


    // UintSet

    struct UintSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(UintSet storage set, uint256 value) internal returns (bool) {
        return _add(set._inner, bytes32(value));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(UintSet storage set, uint256 value) internal returns (bool) {
        return _remove(set._inner, bytes32(value));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(UintSet storage set, uint256 value) internal view returns (bool) {
        return _contains(set._inner, bytes32(value));
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function length(UintSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

   /**
    * @dev Returns the value stored at position `index` in the set. O(1).
    *
    * Note that there are no guarantees on the ordering of values inside the
    * array, and it may change when more values are added or removed.
    *
    * Requirements:
    *
    * - `index` must be strictly less than {length}.
    */
    function at(UintSet storage set, uint256 index) internal view returns (uint256) {
        return uint256(_at(set._inner, index));
    }
}

// File: @openzeppelin/contracts/access/AccessControl.sol



pragma solidity ^0.7.0;




/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context {
    using EnumerableSet for EnumerableSet.AddressSet;
    using Address for address;

    struct RoleData {
        EnumerableSet.AddressSet members;
        bytes32 adminRole;
    }

    mapping (bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view returns (bool) {
        return _roles[role].members.contains(account);
    }

    /**
     * @dev Returns the number of accounts that have `role`. Can be used
     * together with {getRoleMember} to enumerate all bearers of a role.
     */
    function getRoleMemberCount(bytes32 role) public view returns (uint256) {
        return _roles[role].members.length();
    }

    /**
     * @dev Returns one of the accounts that have `role`. `index` must be a
     * value between 0 and {getRoleMemberCount}, non-inclusive.
     *
     * Role bearers are not sorted in any particular way, and their ordering may
     * change at any point.
     *
     * WARNING: When using {getRoleMember} and {getRoleMemberCount}, make sure
     * you perform all queries on the same block. See the following
     * https://forum.openzeppelin.com/t/iterating-over-elements-on-enumerableset-in-openzeppelin-contracts/2296[forum post]
     * for more information.
     */
    function getRoleMember(bytes32 role, uint256 index) public view returns (address) {
        return _roles[role].members.at(index);
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) public virtual {
        require(hasRole(_roles[role].adminRole, _msgSender()), "AccessControl: sender must be an admin to grant");

        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) public virtual {
        require(hasRole(_roles[role].adminRole, _msgSender()), "AccessControl: sender must be an admin to revoke");

        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) public virtual {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        emit RoleAdminChanged(role, _roles[role].adminRole, adminRole);
        _roles[role].adminRole = adminRole;
    }

    function _grantRole(bytes32 role, address account) private {
        if (_roles[role].members.add(account)) {
            emit RoleGranted(role, account, _msgSender());
        }
    }

    function _revokeRole(bytes32 role, address account) private {
        if (_roles[role].members.remove(account)) {
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



pragma solidity ^0.7.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: localhost/contracts/BTCPriceBet.sol


pragma solidity 0.7.6;




/**
 * @dev Players make bets for future BTC price. ETH and own Token can be used for staking
 */
contract BTCPriceBet is AccessControl {
  using SafeMath for uint256;

  bool paused;
    
  uint8 constant public PRIZE_PERCENTAGE = 30;
  uint64 constant private PRICE_MAX = 10e9; //  < 1M
  uint256 constant private INITIAL_BET_DURATION = 10 minutes;
  uint256 constant private BET_ALLOWED_DURATION = 50 minutes;
  bytes32 constant private PRICE_FEED_ROLE = "PRICE_FEED_ROLE";
  bytes32 constant private OWNER_ROLE = "OWNER";
  
  uint256 public betEth;
  uint256 public betTokens;

  uint256 public ongoingRound;
  uint256 public roundStartedAt;
  uint256 public roundDuration;

  address public token;

  struct Bet {
    address account;
    uint256 timestamp;
    uint64 price;
  }
  
  struct HistoryBet {
    bool isEth;
    uint256 timestamp;
    uint64 price;
  }
  
  mapping(address => mapping(address => HistoryBet[])) private betHistory; //  token => (address => HistoryBet[])
  mapping(address => uint64[]) public pricesUsed;   //  token => price[], 0x0 - ETH
  mapping(address => mapping(uint256 => mapping(uint256 => Bet[]))) private betsForRound;   //  token => (round => (price => Bet)), 0x0 - ETH
  mapping(address => mapping(address => uint256)) private lastRoundWithBetMade; //  token => (address => round), 0x0 - ETH
  mapping(address => uint256) public betsNumber;    //  token => bets, 0x0 - ETH

  mapping(address => mapping(address => uint256)) private prizeWithdrawn;   //  token => (address => amount), 0x0 - ETH
  mapping(address => mapping(address => uint256)) private prizeToWithdraw;  //  token => (address => amount), 0x0 - ETH

  mapping(address => mapping(uint256 => uint64[4])) private priceAndWinningPricesForRound;  //  token => uint64[], 0x0 - ETH
  
  mapping(address => uint256) private devFee;   //  token => fee
  
  event BetMade();
  event RoundFinished();

  /***
   * @notice Starts betting round.
   * @dev Contract constructor.
   * @param _token Token address.
   * @param _priceFeed Address, that should be used as price feed.
   */
  constructor(address _token, address _priceFeed) {
    require(_token != address(0), "Wrong token");
    require(_priceFeed != address(0), "Wrong PRICE_FEED_ROLE");

    token = _token;
    betEth = 2e16;  //  0.02 ETH
    betTokens = 10e2; //  1000
    roundDuration = 1 hours;
    roundStartedAt = block.timestamp;
    ongoingRound = 1;
      
    _setupRole(PRICE_FEED_ROLE, _priceFeed);
    _setupRole(OWNER_ROLE, msg.sender);
    _setRoleAdmin(OWNER_ROLE, OWNER_ROLE);
  }
  
  /***
   * @dev Makes bet.
   * @param _token Token address.
   * @param _price Price value.
   */
  function makeBet(address _token, uint64[] calldata _prices) external payable {
    require(!paused, "Paused");
    require(_token == address(0) || _token == token, "Wrong token");
    
    uint256 pricesNumber = _prices.length;
    require(pricesNumber > 0, "Wrong prices num");
    
    require(block.timestamp < roundStartedAt.add(BET_ALLOWED_DURATION), "No more bets");

    if (block.timestamp <= roundStartedAt.add(INITIAL_BET_DURATION)) {
        if (lastRoundWithBetMade[_token][msg.sender] != ongoingRound) {
             lastRoundWithBetMade[_token][msg.sender] = ongoingRound;   
        }
    } else {
      require(lastRoundWithBetMade[_token][msg.sender] == ongoingRound, "Not allowed to bet");
    }

    if (_token == address(0)) {
      require(msg.value == betEth.mul(pricesNumber), "Wrong betEth");
    } else {
      require(msg.value == 0, "Remove eth value");
      IERC20(token).transferFrom(msg.sender, address(this), betTokens.mul(pricesNumber));
    }
    
    for(uint256 i = 0; i < pricesNumber; i ++) {
        uint64 price = _prices[i];
        require((price > 0) && (price < PRICE_MAX), "Wrong price");
        
        if (betsForRound[_token][ongoingRound][price].length == 0) {
          pricesUsed[_token].push(price);
        }
    
        betsForRound[_token][ongoingRound][price].push(Bet(msg.sender, block.timestamp, price));
        betHistory[_token][msg.sender].push(HistoryBet(_token == address(0), block.timestamp, price));
    }
        
    betsNumber[_token] = betsNumber[_token].add(pricesNumber);
    emit BetMade();
  }

  /***
  * @dev Finishes ongoing round.
  * @param _price Price value.
  * @param _winnerPricesETH  Prices for ETH bets.
  * @param _winnerPricesToken  Prices for Token bets.
  */
  function finishRound(uint64 _price, uint64[3] memory _winnerPricesETH, uint64[3] memory _winnerPricesToken) external {
    require(hasRole(PRICE_FEED_ROLE, msg.sender), "Not PRICE_FEED_ROLE");
    require(block.timestamp > roundStartedAt.add(roundDuration), "Not finished yet");

    priceAndWinningPricesForRound[address(0)][ongoingRound][0] = _price;
    priceAndWinningPricesForRound[token][ongoingRound][0] = _price;
    
    uint256 betsEth = betsNumber[address(0)].mul(betEth);
    uint256 betsToken = betsNumber[token].mul(betTokens);

    uint256 prizeEth = betsEth.mul(PRIZE_PERCENTAGE).div(100);
    uint256 prizeToken = betsToken.mul(PRIZE_PERCENTAGE).div(100);
    
    uint256 prizeUsedEth;
    uint256 prizeUsedToken;
    
    for (uint8 i = 0; i < _winnerPricesETH.length; i++) {
      if (betsForRound[address(0)][ongoingRound][_winnerPricesETH[i]].length > 0) {
        address winner = betsForRound[address(0)][ongoingRound][_winnerPricesETH[i]][0].account;
        prizeToWithdraw[address(0)][winner] = prizeToWithdraw[address(0)][winner].add(prizeEth);
        priceAndWinningPricesForRound[address(0)][ongoingRound][i+1] = _winnerPricesETH[i];
        prizeUsedEth = prizeUsedEth.add(prizeEth);
      }

      if (betsForRound[token][ongoingRound][_winnerPricesToken[i]].length > 0) {
        address winner = betsForRound[token][ongoingRound][_winnerPricesToken[i]][0].account;
        prizeToWithdraw[token][winner] = prizeToWithdraw[token][winner].add(prizeToken);
        priceAndWinningPricesForRound[token][ongoingRound][i+1] = _winnerPricesToken[i];
        prizeUsedToken = prizeUsedToken.add(prizeToken);
      }
    }
    
    //  if only 1 player, he gets all prizes.
    if (betsEth > 0) {
      if (prizeUsedEth == prizeEth) {
        address winner = betsForRound[address(0)][ongoingRound][_winnerPricesETH[0]][0].account;
        prizeToWithdraw[address(0)][winner] = prizeToWithdraw[address(0)][winner].add(prizeEth.mul(2));
        prizeUsedEth = prizeUsedEth.mul(3);
      }
      devFee[address(0)] = devFee[address(0)].add(betsEth.sub(prizeUsedEth));
    }
    
    if (betsToken > 0) {
      if (prizeUsedToken == prizeToken) {
        address winner = betsForRound[token][ongoingRound][_winnerPricesToken[0]][0].account;
        prizeToWithdraw[token][winner] = prizeToWithdraw[token][winner].add(prizeToken.mul(2));
        prizeUsedToken = prizeUsedToken.mul(3);
      }
      devFee[token] = devFee[token].add(betsToken.sub(prizeUsedToken));  
    }

    //  clear, update data
    delete pricesUsed[address(0)];
    delete pricesUsed[token];
    
    delete betsNumber[address(0)];
    delete betsNumber[token];
    
    ongoingRound = ongoingRound.add(1);
    roundStartedAt = block.timestamp;
    
    emit RoundFinished();
  }


  /***
  * @dev Withdraws prize.
  * @param _token Token address.
  */
  function withdrawPrize(address _token) external {
    uint256 prize = getPrizeToWithdraw(_token);
    require(prize > 0, "No prize");

    if (_token == address(0)) {
      msg.sender.transfer(prize);
    } else {
      require(IERC20(_token).transfer(msg.sender, prize), "Transfer failed");
    }

    delete prizeToWithdraw[_token][msg.sender];
    prizeWithdrawn[_token][msg.sender] = prizeWithdrawn[_token][msg.sender].add(prize);
  }
  
  /***
  * @dev Gets pending prize for sender.
  * @param _token Token address.
  * @return Prize amount.
  */
  function getPrizeToWithdraw(address _token) public view returns(uint256) {
    require(_token == address(0) || _token == token, "Wrong token");
    
    return prizeToWithdraw[_token][msg.sender];
  }
  
  /***
  * @dev Gets prize withdrawn amount.
  * @param _token Token address.
  * @return Prize withdrawn.
  */
  function getPrizeWithdrawn(address _token) external view returns(uint256) {
    require(_token == address(0) || _token == token, "Wrong token");

    return prizeWithdrawn[_token][msg.sender];
  }

  /***
  * @dev Withdraws dev fee.
  * @param _token Token address.
  */
  function withdrawDevFee(address _token) external {
    uint256 fee = getDevFee(_token);
    require(fee > 0, "No fee");
    delete devFee[_token];

    if (_token == address(0)) {
      msg.sender.transfer(fee);
    } else {
      require(IERC20(_token).transfer(msg.sender, fee), "Transfer failed");
    }
  }

  /***
  * @dev Gets dev fee.
  * @param _token Token address.
  * @return dev fee.
  */
  function getDevFee(address _token) public view returns(uint256) {
    require(hasRole(OWNER_ROLE, msg.sender), "Not OWNER_ROLE");
    require(_token == address(0) || _token == token, "Wrong token");
    
    return devFee[_token];
  }

  /***
  * @dev Prices used for ongoing round.
  * @param _token Token address.
  * @return Prices used.
  */
  function getPricesUsed(address _token) external view returns (uint64[] memory) {
    require(_token == address(0) || _token == token, "Wrong token");

    return pricesUsed[_token];
  }

  /***
  * @dev Gets BTC price and winning prices for round.
  * @param _token Token address.
  * @param _round Round index.
  * @return prices Prices.
  */
  function getPriceAndWinningPricesForRound(address _token, uint256 _round) external view returns(uint64[4] memory) {
    require(_token == address(0) || _token == token, "Wrong token");
    require(_round < ongoingRound, "Wrong round");

    return priceAndWinningPricesForRound[_token][_round];
  }
  
  /***
   * @dev Gets BTC price and winning prices for round.
   * @param _token Token address.
   * @param _round Round index.
   * @param _price Price.
   * @return Bet properties list (accounts[], timestamps[], prices[]).
   */
  function getBetsForRoundForPrice(address _token, uint256 _round, uint64 _price) view external returns(address[] memory, uint256[] memory, uint64[] memory) {
    require(_token == address(0) || _token == token, "Wrong token");
    require((_price > 0) && (_price < PRICE_MAX), "Wrong price");
    
    Bet[] storage bets = betsForRound[_token][_round][_price];
    uint256 betsLength = bets.length;
      
    address[] memory accounts = new address[](betsLength);
    uint256[] memory timestamps = new uint256[](betsLength);
    uint64[] memory prices = new uint64[](betsLength);
      
    for (uint256 i = 0; i < betsLength; i ++) {
      accounts[i] = bets[i].account;
      timestamps[i] = bets[i].timestamp;
      prices[i] = bets[i].price;
    }
      
    return (accounts, timestamps, prices);
  }
  
  /***
   * @dev Gets last round, when sender participated.
   * @param _token Token address.
   * @return Round idx.
   */
  function getLastRoundWithBetMade(address _token) view external returns (uint256) {
      require(_token == address(0) || _token == token, "Wrong token");
      
      return lastRoundWithBetMade[_token][msg.sender];
  }
  
  /***
   * @dev Gets bets count for sender.
   * @param _token Token address.
   * @return Bets count idx.
   */
  function getBetHistoryCountFor(address _token) view external returns (uint256) {
    return betHistory[_token][msg.sender].length;   
  }
  
  /***
   * @dev Gets bet history count for sender.
   * @param _token Token address.
   * @param _idxStart Index to start.
   * @param _idxEnd Index to end.
   * @return Bet list.
   */
  function getBetHistoryFor(address _token, uint256 _idxStart, uint256 _idxEnd) view external returns (bool[] memory, uint256[] memory, uint64[] memory) {
    HistoryBet[] storage bets = betHistory[_token][msg.sender];
    uint256 betsLength = bets.length;
    
    if (betsLength == 0) {
        return(new bool[](0), new uint256[](0), new uint64[](0));
    }
    
    uint256 start = 0;
    uint256 end = betsLength.sub(1);
    
    if (_idxEnd > 0) {
        require(_idxStart < _idxEnd && _idxEnd < betsLength, "Wrong idxs");
        
        start = _idxStart;
        end = _idxEnd;
    }
    
    uint256 arrLength = end.sub(start).add(1);
      
    bool[] memory isEth = new bool[](arrLength);
    uint256[] memory timestamps = new uint256[](arrLength);
    uint64[] memory prices = new uint64[](arrLength);
          
    for (uint256 i = start; i <= end; i ++) {
      isEth[i] = bets[i].isEth;
      timestamps[i] = bets[i].timestamp;
      prices[i] = bets[i].price;
    }
          
    return (isEth, timestamps, prices);   
  }


  /***
  * @dev Updates bet value for eth.
  * @param _isToken Update bet for token or eth.
  * @param _value Bet value to be used.
  */
  function updateBet(bool _isToken, uint256 _value) external {
    require(hasRole(OWNER_ROLE, msg.sender), "Not OWNER_ROLE");
    require(_value > 0, "Wrong bet value");

    (_isToken) ? betTokens = _value : betEth = _value;
  }

  /***
  * @dev Updates round duration.
  * @param _duration Duration value to be used.
  */
  function updateRoundDuration(uint256 _duration) external {
    require(hasRole(OWNER_ROLE, msg.sender), "Not OWNER_ROLE");
    require(_duration > 0, "Wrong duration");

    roundDuration = _duration;
  }

  /***
  * @dev Pauses / unpauses betting functional.
  * @param _isPaused Set paused value.
  */
  function setPaused(bool _isPaused) external {
    require(hasRole(OWNER_ROLE, msg.sender), "Not OWNER_ROLE");

    paused = _isPaused;
  }

  /***
  * @dev Kills Smart Contract.
  */
  function kill() external {
    require(hasRole(OWNER_ROLE, msg.sender), "Not OWNER_ROLE");
    selfdestruct(payable(msg.sender));
  }
}